


Дополнительные возможности по настройке графиков

Аргумент команды **typeset** обрабатывается системой Maple, поэтому фактическая подпись по оси может отличаться от того, что было подано на вход. (По умолчанию полиномы выводятся в порядке убывания степеней отдельных термов.)

```
> plot(sin(x)/(1+x^2), x=-8..8,  
      labels=[typeset(x), typeset(sin(x)/(1+x^2))]);
```




Длинные опции лучше вынести из команды **plot**,

```
> LL:= [typeset(x), typeset(sin(x)/(1+x^2))];
```

$$LL := \left[\text{typeset}(x), \text{typeset}\left(\frac{\sin(x)}{x^2 + 1}\right) \right]$$

тогда сама команда рисования будет выглядеть не так жутко.

```
> plot(sin(x)/(1+x^2), x=-8..8, labels=LL);
```



Над математическими выражениями можно выполнять любые команды Maple.


В данном случае мы отсортировали знаменатель в порядке возрастания степеней.

```
> LL:= [typeset(x), typeset(applyop(sort,2,sin(x)/(1+x^2),x,
ascending))] ;
```

$$LL := \left[\text{typeset}(x), \text{typeset}\left(\frac{\sin(x)}{1+x^2}\right) \right]$$

Таким образом, например, можно обеспечить желаемый вид знаменателя дроби

```
> plot(sin(x)/(1+x^2),x=-8..8,labels=LL) ;
```




Сами зависимости, которые мы хотели бы изобразить, тоже лучше хранить отдельно от команды **plot**.

```
> F:=sqrt(x)/(1+x*ln(x)) ;
```

$$F := \frac{\sqrt{x}}{1+x \ln(x)}$$


Это позволяет неоднократно ссылаться на внешние объекты.

```
> plot(F,x=0..5,labels=[typeset(x), typeset(F)]) ;
```



Для вывода меток может использоваться свой собственный шрифт. Он задаётся при помощи опции *labelfont*. Параметры этой опции такие же, как и у опции *titlefont*.


```
> plot(F,x=0..5,labels=[typeset(x),typeset(F)],
      labelfont=["Times","Roman",12]);
```



Maple позволяет выполнить сразу несколько команд, объединив их в группу.

```
> Pol:=2*x^4-5*x^3-7*x^2+12*x-4;
plot(Pol,x=-2..4,y=-20..20,labels=[typeset(x),typeset(Pol)]);
```


$$Pol := 2x^4 - 5x^3 - 7x^2 + 12x - 4$$



Длинные горизонтальные метки по оси ординат смотрятся ужасно. Их следует развернуть с помощью опции *labeldirections*. Значением этой

опции является список из двух элементов. Первый элемент указывает направление вывода подписи по оси абсцисс. Второй - по оси ординат. По умолчанию оба элемента имеют значение *horizontal*. В целях безопасности константы *horizontal* и *vertical* следует заключать в двойные кавычки. Однако, если во время сеанса работы в системе компьютерной алгебры вы не использовали переменные с такими именами, то кавычки можно опустить.

```
> plot(Pol,x=-2..4,y=-20..20,labels=[typeset(x),typeset(Pol)],  
      labeldirections=["horizontal","vertical"]);
```




Если введённое ранее выражение нам больше не нужно, то переменную, в которой оно хранилось, можно использовать для других целей.

```
> F:=exp(-x^2)+2*exp(-(x-3)^2);
```

$$F := e^{-x^2} + 2e^{-(x-3)^2}$$


Для удобства дальнейшей работы с графиком на изображение можно наложить масштабную сетку. Это делается с помощью опции *gridlines*. По умолчанию её значение равно *false*, поэтому сетка не рисуется. Присвоив этой опции значение *true* или же просто написав *gridlines*, получаем результат, показанный ниже.

```
> plot(F,x=-2..5,labels=[typeset(x),typeset(F)],  
      labelfont=["Times","Roman",12],  
      labeldirections=["horizontal","vertical"],  
      gridlines=true);
```




В каких-то экзотических случаях может оказаться полезной возможность закрасить область между графиком функции и осью абсцисс некоторым цветом. Такая возможность реализуется с помощью опции *filled*. Её синтаксис и значение по умолчанию такие же как и у *gridlines*. Присвоив *filled* значение *true*, предыдущий рисунок преобразуется к следующему виду:

```
> plot(F,x=-2..5,filled=true);
```




Как уже отмечалось ранее, при построении графиков Maple пытается автоматически подбирать масштаб изображения по оси ординат. Такая особенность может приводить к нежелательным искажениям при рисовании геометрических объектов. Например, при попытке изобразить полуокружность единичного радиуса на экране появится половина эллипса.

```
> plot(sqrt(1-x^2),x=-1..1);
```



Для предотвращения подобных эффектов следует принудительно задать одинаковые масштабы по обеим координатным осям. Именно для этого предназначена опция *scaling*. По умолчанию её значение равно *unconstrained*, то есть за выбор масштаба отвечает система компьютерной алгебры. Если же мы хотим увидеть на плоскости привычные, а не искажённые кривые и фигуры, то в качестве значения нужно указать *constrained*.

```
> plot(sqrt(1-x^2), x=-1..1, scaling=constrained);
```




Если системе Maple не давать никаких указаний, то все отображаемые линии будут сплошными. К счастью такое поведение, принятое по умолчанию, можно изменить при помощи опции *linestyle*. В старых версиях Maple для изменения стиля линии этой опции присваивались в качестве значений целые числа от 1 до 7. Такой подход удобен с точки зрения программирования, однако он не слишком

наглядно. Сейчас в режиме диалога с системой более привычно использовать мнемонические имена из следующего списка:


1. *solid*
2. *dot*
3. *dash*
4. *dashdot*
5. *longdash*
6. *spacedash*
7. *spacedot*

```
> plot(sqrt(1-x^2), x=-1..1, scaling=constrained, linestyle=dot);
```



При рисовании геометрических объектов вполне можно обойтись и без координатных осей. Их вид настраивается с помощью опции *axes*. Если указать значение *none*, то координатные оси рисоваться **не будут**.

```
> plot(sqrt(1-x^2), x=-1..1, scaling=constrained, axes=none);
```



Кому-то, наоборот, могут понравиться масштабные метки по краям рамки, в которую заключён рисунок. В таком случае следует написать *axes=boxed*.

```
> plot(sqrt(1-x^2), x=-1..1, scaling=constrained, axes=boxed);
```

